
Elliptic Curve Demo

Part I: Elliptic curves over Q

Graphs of some elliptic curves

E = EllipticCurve([-2,0]); E
       Elliptic Curve defined by y^2  = x^3 - 2*x over Rational Field
show(E)

       

show(plot(E, thickness=3, hue=0.7))

       

v = []
for E in cremona_optimal_curves(range(40)):
    v.append(plot(E, thickness=5, rgbcolor=(random(),random(),random())))
n = len(v)
m = floor(sqrt(n))
show(graphics_array(v, m, m))

y x 2 = x3 − 2

Elliptic curve group law examples

E = EllipticCurve([-5, 4]); E
       Elliptic Curve defined by y^2  = x^3 - 5*x + 4 over Rational Field
G = plot(E, thickness=2)
G.show(ymin=-3,ymax=3,figsize=[2,4])

       

P = E(1,0); Q = E(0,2); R = P+Q
print R

       (3 : 4 : 1)
def pnt(z,clr):
    return plot(z, pointsize=200, rgbcolor=clr)
show(G + pnt(P,(1,0,0)) + pnt(Q,(0,1,0)) + \
       pnt(R,(0.5,0.5,0)) + pnt(-R,(0,0,1)) + \
     line([(-1,4), (4,-6)],rgbcolor=(0,0,1),thickness=2) + \
     line([(3,-5), (3, 5)], rgbcolor=(0,0,1),thickness=2) + \
     text("(0,2) + (1,0) = (3,4)", (2.7,4.5), \
           fontsize=14, horizontal_alignment='right'), \
    figsize=[3,6], ymin=-5,ymax=4)

       

3d Graph of Rational Points

# This is my favorite elliptic curve
E = EllipticCurve('389a')
       
show(plot(E))

t = Tachyon(xres=600, yres=500, \
          camera_center=(4,7,4), look_at=(1,0,0), raydepth=4)
t.light((10,3,2), 1, (1,1,1))
t.light((10,-3,2), 1, (1,1,1))
t.texture('black', color=(0,0,0))
t.texture('red', color=(1,0,0))
t.texture('grey', ambient=1,diffuse=1, specular=0, opacity=1, color=(0.8,.8,0.8))
t.plane((0,0,0),(0,0,1),'white')
t.cylinder((0,0,0),(1,0,0),.01,'black')
t.cylinder((0,0,0),(0,1,0),.01,'black')
E = EllipticCurve('389a')
G = E.gens()
n = 6
for i in range(-n,n+1):  
  for j in range(-n,n+1):
    if i == 0 and j == 0: continue
    Q = i*G[0] + j*G[1]
    t.texture('r%s'%i,color=(float(i)/n, 0, float(j)/n))
    t.sphere((Q[0], -Q[1], .02), .04, 'r%s'%i)
          
t.save('sage.png')    # long time, seconds

# The rational points get quite large.
P = E(0,0)
show([n*P for n in range(15)])

Part II: Elliptic curves modulo p

Graph of Set of Points modulo p

E = EllipticCurve(GF(11), [-5,4]); E
       Elliptic Curve defined by y^2  = x^3 + 6*x + 4 over Finite Field of size 11
show(plot(E,pointsize=200,hue=.7))

       

E = EllipticCurve(GF(2011), [-5,4]); E
       Elliptic Curve defined by y^2  = x^3 + 2006*x + 4 over Finite Field of size 2011
show(plot(E))

Arithmetic on elliptic curve modulo p

E = EllipticCurve(GF(19), [-5, 4])
P = E(1,0); Q = E(0,2)
       
P +  Q
       (3 : 4 : 1)
7*Q
       (16 : 7 : 1)
P + Q + Q  + Q
       (15 : 6 : 1)

Diffie Hellman key exchange on elliptic curve modulo 
p

p = 785963102379428822376694789446897396207498568951
show(p.str(16))

       

E = EllipticCurve(GF(p), [317689081251325503476317476413827693272746955927, \
   79052896607878758718120572025718535432100651934])
E

       

Elliptic Curve defined by y^2  = x^3 +
317689081251325503476317476413827693272746955927*x +
79052896607878758718120572025718535432100651934 over Finite Field of size
785963102379428822376694789446897396207498568951

time s = E.cardinality()
s

       
785963102379428822376693024881714957612686157429
CPU time: 0.00 s,  Wall time: 0.01 s

is_prime(s)
       True
B = E(771507216262649826170648268565579889907769254176, \
      390157510246556628525279459266514995562533196655)
       

89abcdef012345672718281831415926141424f7 

%time
# Do Diffie-Hellman with this
n = randint(2,s)
m = randint(2,s)
Sm = m*(n*B)
Sn = n*(m*B)
       CPU time: 0.10 s,  Wall time: 0.12 s
Sm

       
(153470721398328205142504643349771892468236940170 :
132434024879389525183838911880841584368869098553 : 1)

Sn

       
(153470721398328205142504643349771892468236940170 :
132434024879389525183838911880841584368869098553 : 1)

       


